
	  function f_IMS_CommunicationWaiting_CancelCall(INVITE_Request p_InviteRequest) runs on IMS_PTC
  {
    var IMS_DATA_RSP v_IMS_DATA_RSP;
    var template (value) MessageHeader v_MessageHeader_Ack;
    var template (value) MessageHeader v_MessageHeader_Cancel := f_IMS_CancelRequest_MessageHeaderTX(p_InviteRequest);
    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();
    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);      /* @sic R5s150908 change 4 sic@ */
    var SipUrl v_RequestUrl := p_InviteRequest.requestLine.requestUri;
    var boolean v_WaitingFor200 := true;
    var boolean v_WaitingFor487 := true;

    IMS_Client.send(cas_IMS_Cancel_Request(cs_IMS_RoutingInfo(v_Protocol), cs_CANCEL_Request(v_RequestUrl ContactUrl, v_MessageHeader_Cancel)));
    
    while (v_WaitingFor200 or v_WaitingFor487) {
      alt { // @sic R5-125773, R5s140350: interleave is not allowed for altsteps sic@
        // The UE acknowledges the CANCEL with 200 OK:
        [v_WaitingFor200] a_IMS_ReceiveResponse(v_MessageHeader_Cancel,
                                                v_Protocol,
                                                cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Cancel)),
                                                v_IMS_DATA_RSP) {
          v_WaitingFor200 := false;
        }
        // The UE responds to INVITE with a 487 Request Terminated final response after transaction was terminated
        [v_WaitingFor487] a_IMS_ReceiveResponse(p_InviteRequest.msgHeader,
                                                v_Protocol,
                                                cr_Response(c_statusLine487, f_IMS_InviteResponse_4XX_MessageHeaderRX(p_InviteRequest.msgHeader)),       /* @sic R5s140178 sic@ */
                                                v_IMS_DATA_RSP) {
          v_WaitingFor487 := false;
        }
      }
    }
    // Send ACK. The SS acknowledges the receipt of 200 OK for INVITE.
    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(p_InviteRequest, c_statusLine487);   /* @sic R5s140092 change 1.3; R5s140178 sic@ */
    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_ACK_Request(v_RequestUrl ContactUrl, v_MessageHeader_Ack)));
  }



